ECE 602 Lecture Notes: Helicopter Example

Consider the "high performance helicopter" example in DP11.4, p.822 of Dorf & Bishop's, 11th ed.

The equations of motion are

$$\frac{d^2x}{dt^2} = g\theta - \alpha_2\frac{d\theta}{dt} - \sigma_2\frac{dx}{dt} + g\delta \tag{1}$$

$$\frac{d^2\theta}{dt^2} = -\sigma_1\frac{d\theta}{dt} - \alpha_1\frac{dx}{dt} + n\delta \tag{2}$$

where $x(t)$ is the horizontal displacement, $\theta(t)$ is the pitch angle, and $\delta(t)$ is the input, which is rotor thrust angle.

The first step is to rewrite this in state space. We obtain

$$\begin{bmatrix} \dot{x} \\ \ddot{x} \\ \dot{\theta} \\ \ddot{\theta} \end{bmatrix} = \begin{bmatrix} 0 & 1 & 0 & 0 \\ 0 & -\sigma_2 & g & -\alpha_2 \\ 0 & 0 & 0 & 1 \\ 0 & -\alpha_1 & 0 & -\sigma_1 \end{bmatrix} \begin{bmatrix} x \\ x \\ \theta \\ \dot{\theta} \end{bmatrix} + \begin{bmatrix} 0 \\ g \\ 0 \\ n \end{bmatrix} \delta \tag{3}$$

We will use Maple to help with the calculations. The first two commands clear the workspace and include the linear algebra commands so that we can use them.

```
> restart:
> with(linalg):
```

Next, we input the matrices:

```
> A := matrix(4,4,[0,1,0,0,0,-sigma2,g,-alpha2,0,0,0,1,0,-alpha1,0,-sigma1]);

                   [0       1        0        0    ]
                   [                               ]
                   [0    -sigma2     g    -alpha2]
              A := [                               ]
                   [0       0        0        1    ]
                   [                               ]
                   [0    -alpha1     0    -sigma1]
```

```
> B := matrix(4,1,[0,g,0,n]);

                                    [0]
                                    [ ]
                                    [g]
                               B := [ ]
                                    [0]
                                    [ ]
                                    [n]

> C := matrix(1,4,[0,0,1,0]);

                         C := [0    0    1    0]
```

Next, we find the transfer function from $\delta(s)$ to $\theta(s)$.

```
> eye4 := matrix(4,4,[1,0,0,0,0,1,0,0,0,0,1,0,0,0,0,1]);

                              [1    0    0    0]
                              [                ]
                              [0    1    0    0]
                      eye4 := [                ]
                              [0    0    1    0]
                              [                ]
                              [0    0    0    1]

> sImA := evalm(s*eye4-A);

                    [s        -1         0          0     ]
                    [                                     ]
                    [0    s + sigma2    -g      alpha2    ]
            sImA := [                                     ]
                    [0        0          s         -1     ]
                    [                                     ]
                    [0      alpha1       0     s + sigma1]
```

```
> G := collect(simplify(multiply(C,inverse(sImA),B)),s);

> G :=

        [                                        3                        2
        [(-alpha1 g + n s + n sigma2)/(s  + (sigma2 + sigma1) s

                                                         ]
         + (-alpha2 alpha1 + sigma2 sigma1) s + alpha1 g)]
```

The resulting expression for $\hat{g}(s)$ is

$$\hat{g}(s) = \frac{ns + n\sigma_2 - \alpha_1 g}{s^3 + (\sigma_2 + \sigma_1)s^2 + (-\alpha_2\alpha_1 + \sigma_2\sigma_1)s + \alpha_1 g}. \tag{4}$$

Next, we calculate the controllability and observability matrices and check their ranks.

```
> Cee := concat(B,multiply(A,B),multiply(A,A,B),multiply(A,A,A,B));

  Cee :=

        [0 , g , -sigma2 g - alpha2 n , %4 g + %3 n]

        [g , -sigma2 g - alpha2 n , %4 g + %3 n ,

        (-%4 sigma2 - %3 alpha1) g

         + (-%4 alpha2 - sigma2 g - %3 sigma1) n]

        [0 , n , -alpha1 g - sigma1 n , %2 g + %1 n]

        [n , -alpha1 g - sigma1 n , %2 g + %1 n ,

        (-%2 sigma2 - %1 alpha1) g

         + (-%2 alpha2 - alpha1 g - %1 sigma1) n]

                                    2
  %1 := alpha2 alpha1 + sigma1

  %2 := alpha1 sigma2 + sigma1 alpha1

  %3 := sigma2 alpha2 + g + alpha2 sigma1
```

```
               2
  %4 := sigma2  + alpha2 alpha1

> rank(Cee);

                                    4


> Ohh := stackmatrix(C,multiply(C,A),multiply(C,A,A),multiply(C,A,A,A));

  Ohh :=

        [0 , 0 , 1 , 0]

        [0 , 0 , 0 , 1]

        [0 , -alpha1 , 0 , -sigma1]

        [
        [0 , alpha1 sigma2 + sigma1 alpha1 , -alpha1 g ,

                                  2]
        alpha2 alpha1 + sigma1 ]

> rank(Ohh);

                                    3
```

Since the system is not observable, we construct an equivalence transformation by selecting three linearly independent rows of $\mathcal{O}$ and augmenting them to a basis.

```
> P := matrix(4,4,[0,0,1,0,0,0,0,1,0,-alpha1,0,-sigma1,1,0,0,0]);

                      [0        0        1        0    ]
                      [                                ]
                      [0        0        0        1    ]
                 P := [                                ]
                      [0    -alpha1      0    -sigma1]
                      [                                ]
                      [1        0        0        0    ]
```

```
> Q := inverse(P);

                   [0       0            0         1]
                   [                                ]
                   [      sigma1         1          ]
                   [0    - ------    - ------      0]
              Q := [      alpha1       alpha1       ]
                   [                                ]
                   [1       0            0         0]
                   [                                ]
                   [0       1            0         0]
```

Now we can transform the system to Kalman observability form.

```
> tildeA := simplify(multiply(P,A,Q));

  tildeA :=

        [0 , 1 , 0 , 0]

        [0 , 0 , 1 , 0]

        [-alpha1 g , -sigma2 sigma1 + alpha2 alpha1 ,

        -sigma2 - sigma1 , 0]

        [      sigma1       1       ]
        [0 , - ------ , - ------ , 0]
        [      alpha1     alpha1    ]

> tildeB := simplify(multiply(P,B));

                           [         0          ]
                           [                    ]
                           [         n          ]
                 tildeB := [                    ]
                           [-alpha1 g - sigma1 n]
                           [                    ]
                           [         0          ]

> tildeC := simplify(multiply(C,Q));
```

```
tildeC := [1    0    0    0]
```

Note that the format of the matrices matches that in Theorem 6.O.6 (p. 162) of the text for ECE602 with

$$\bar{A}_o = \begin{bmatrix} 0 & 1 & 0 \\ 0 & 0 & 1 \\ -\alpha_1 g & (\alpha_2\alpha_1 - \sigma_2\sigma_1) & -(\sigma_2 + \sigma_1) \end{bmatrix} \tag{5}$$

$$\bar{A}_{12} = \begin{bmatrix} 0 & -\sigma_1/\alpha_1 & -1/\alpha_1 \end{bmatrix} \tag{6}$$

$$\bar{B}_o = \begin{bmatrix} 0 \\ n \\ -(\alpha_1 g + \sigma_1 n) \end{bmatrix} \tag{7}$$

$$\bar{C}_o = \begin{bmatrix} 1 & 0 & 0 \end{bmatrix} \tag{8}$$

Of course we must not forget that the vector variable $\tilde{x}$ is not the same as the vector x.

```
> tildex := multiply(P,matrix(4,1,[x1,x2,x3,x4]));

                    [          x3          ]
                    [                      ]
                    [          x4          ]
          tildex := [                      ]
                    [-alpha1 x2 - sigma1 x4]
                    [                      ]
                    [          x1          ]
```

We see that the unobservable mode is the horizontal displacement. Luckily, one usually knows where the flight began.